Evaluating econometric inference by Monte Carlo simulation;
foundations , pedagogy, methodology, presentation

by ***Jan F. Kiviet***
(University of Amsterdam, Tinbergen Institute)

UvA ECONOMETRICS

Seminar at **Cass Business School, City University**
London, 26 October 2007

Overview of talk (MC=Monte Carlo simulation)

- A basic illustration of MC
  - OLS, TSLS inference in a dynamic simultaneous model
  - pedagogic tool for teaching econometric theory
- Some remarks on foundation of MC inference
  - clarifying notation
  - MC inference is asymptotic inference
- Rating competing inference techniques by MC
  - MC methodology, illustrated by
  - alternative estimators for dynamic panel data models
- Graphical presentation of MC results
  - using 2D, 3D, or xD in animations

Illustration of MC for teaching purposes

Model with two structural equations (t=1,...,T):

$Y_t = C_t + I_t + G_t$ (budget restriction)

$C_t = \beta_1 + \beta_2 Y_t + \beta_3 C_{t-1} + \varepsilon_t$ (consumption function)

Both investment and government spending are exogenous.
The two reduced form equations are:

$$C_t = [\beta_1 + \beta_2(I_t + G_t) + \beta_3 C_{t-1} + \varepsilon_t]/(1- \beta_2)$$
$$= \gamma_1 + \gamma_2 I_t + \gamma_3 G_t + \gamma_4 C_{t-1} + \varepsilon_t/(1- \beta_2) \qquad \gamma_4 = \beta_3/(1- \beta_2)$$

$$Y_t = [\beta_1 + I_t + G_t + \beta_3 C_{t-1} + \varepsilon_t]/(1- \beta_2)$$

We take: $\varepsilon_t \sim IID(0,\sigma^2)$, and $\beta_1 = 0$; $\beta_2 = 0.3$; $\beta_3 = 0.6$; $\sigma = 0.4$.

Structural form relationship:

$$C_t = 0 + 0.3Y_t + 0.6C_{t-1} + 0.4 \times IID(0,1)$$

Reduced form relationship:

$$C_t = 0 + 0.429(I_t + G_t) + 0.857C_{t-1} + 0.571 \times IID(0,1)$$

We generate artificial (stationary) data:

$I_t = 20 + 0.8\sin(t/4.5)$ has cycle of about 28 quarters

$G_t = 10 - 0.2\sin((t-8)/4.5)$ same (negative) cycle 8 quarters later

$C_1 = 90$

$\varepsilon_t = \sigma(8 - v_t)/4$ with IID $v_t \sim \chi^2(8)$, hence ε_t is skew to the left.

We analyze by Monte Carlo methods OLS and TSLS coefficient estimates, estimators of their standard errors, and the rejection probability of particular test statistics under a valid or invalid null hypothesis, and using valid or invalid inference procedures.

Econometric issues:
- simultaneity
- dynamics
- non-normality
- finite sample

EViews program:
Some particulars of the programming language:
- *one statement/operation per line*
- *text after ‘ is comment*
- *name of a scalar constant starts with !*


```
'========= MC illustration: OLS and 2SLS estimation and inference in stylized (macro) model with stationary data
!T=80                               ' sample size
workfile C:\JFKdocuments\MCillustration.wf1 u 1 !T    ' workfile with !T observations
genr t = @trend(0)                  ' linear trend
genr Inv = 20 + 0.8*sin(t/4.5)      ' Investments (exogenous) with cycle of about 7x4 quarters (2pi x 4.5 = 28)
genr Gov = 10 - 0.2*sin((t-8)/4.5)  ' Government spending (exogenous, 2 years lagged anti-cyclic)
!beta1 = 0                          ' intercept of Consumption function
!beta2 = 0.3                        ' coefficient of Income in Consumption function
!beta3 = 0.6                        ' coefficient of lagged Consumption
!sigma = 0.4                        ' standard deviation of structural disturbance
smpl 1 1
genr Cons = 90                      ' Consumption in observation 1
smpl 2 !T                           ' estimation sample will have !T-1 observations
!R = 10000                          ' number of simulation replications
matrix (!R,6) simres                ' a !Rx6 matrix to store main simulation results
rndseed 26102007                    ' allows to re-obtain the same random drawings

for !rep=1 to !R
  genr eps = !sigma*(8 - @rchisq(8))/4                        ' IID(0, !sigma^2) structural disturbances, though skew
  genr Cons = (!beta1 + !beta2*(Inv + Gov) + !beta3*Cons(-1) + eps)/(1 - !beta2)    ' Cons obtained by reduced form
  genr Inc = (!beta1 + Inv + Gov + !beta3*Cons(-1) + eps)/(1 - !beta2)              ' Income obtained by reduced form
  equation eq1.ls Cons C Inv Gov Cons(-1)                     ' OLS estimation of reduced form
  simres(!rep,1) = eq1.@coefs(4)                              ' b4rOLS = OLS coefficient for lagged Cons in reduced form
  simres(!rep,2) = eq1.@stderrs(4)                            ' seb4rOLS = std.error of b4rOLS
  equation eq2.ls Cons C Inc Cons(-1)                         ' OLS estimation of structural form
  simres(!rep,3) = eq2.@coefs(2)                              ' b2OLS = OLS coefficient for Income
  simres(!rep,4) = eq2.@stderrs(2)                            ' seb2OLS = (alleged) std.error of b2OLS
  equation eq3.tsls Cons C Inc Cons(-1) @ C Gov Inv Cons(-1)                ' TSLS estimation of structural form
  simres(!rep,5) = eq3.@coefs(2)                              ' b2TSLS = TSLS coefficient for Income
  simres(!rep,6) = eq3.@stderrs(2)                            ' seb2TSLS = std.error b2TSLS
next

simres.write C:\JFKdocuments\MCillustration.txt           ' converting simulation results
workfile C:\JFKdocuments\MCillustrationsim.wf1 u 1 !R    ' to a workfile with !R observations and variables with the following
read C:\JFKdocuments\MCillustration.txt b4rOLS seb4rOLS b2OLS seb2OLS b2TSLS seb2TSLS        ' variable names

!cv = @qnorm(0.975)           ' standard normal critical value at one-tailed significance level of 2.5%
genr rejtrueb4rOLS = abs((b4rOLS - !beta3/(1-!beta2))/seb4rOLS) > !cv              ' reject true H0 on 4th red.form coef.
genr rejfalseb4rOLS = abs((b4rOLS - !beta3/(1-!beta2) + 0.1)/seb4rOLS) > !cv       ' rej. false H0 on this 4th r.f. coef.
genr rejtrueb2OLS = abs((b2OLS - !beta2)/seb2OLS) > !cv              ' rej. true H0 on !beta2 under false maintained hyp.
genr rejtrueb2TSLS = abs((b2TSLS - !beta2)/seb2TSLS) > !cv           ' rej. true H0 on !beta2 under valid maintained hyp.
genr rejfalseb2OLS = abs((b2OLS - !beta2 - 0.2)/seb2OLS) > !cv       ' rej. false H0 on !beta2 under false maintained hyp.
genr rejfalseb2TSLS = abs((b2TSLS - !beta2 - 0.2)/seb2TSLS) > !cv   ' rej. false H0 on !beta2 under valid maintained hyp.
```

Some illustrating results obtained in the **final replication**:





EPS is skew to the left indeed

© JFK 2007 7

EQ1, OLS reduced form



B4ROLS = OLS est. of γ_4

$\left[\vec{V}(\hat{\gamma}_4)\right]^{1/2}$ $\vec{E}(\hat{\gamma}_4)$

0.857

true values: 0 0.429 0.429 0.857

Dependent Variable: CONS
Method: Least Squares
Sample: 2 80
Included observations: 79

Variable	Coefficient	Std. Error	t-Statistic	Prob.
C	16.19056	10.07448	1.607086	0.1122
INV	0.400537	0.128954	3.106045	0.0027
GOV	-0.547524	0.592566	-0.923989	0.3585
CONS(-1)	0.792380	0.072356	10.95114	0.0000

R-squared	0.757848	Mean dependent var	90.34697
Adjusted R-squared	0.748162	S.D. dependent var	1.166393
S.E. of regression	0.585336	Akaike info criterion	1.816044
Sum squared resid	25.69636	Schwarz criterion	1.936016
Log likelihood	-67.73376	F-statistic	78.24110
Durbin-Watson stat	1.746343	Prob(F-statistic)	0.000000

0.571

Breusch-Godfrey Serial Correlation LM Test: F-statistic=1.430703 Prob. F(2,73)=0.245766
White Heteroskedasticity Test: F-statistic=1.814925 Prob. F(8,70)=0.088683

$\vec{E}\left\{\left[\hat{V}(\hat{\gamma}_4)\right]^{1/2}\right\}$



With 99.5% probability B4ROLS has bias in CI: (0.779 - 0.857) ± 3 x 0.086/100 = [-0.081 , -0.076]

confidence interval

$\left\{\vec{V}\left[\vec{E}(\hat{\gamma}_4)\right]\right\}^{1/2}$

© JFK 2007 8

EQ2, OLS structural form

$\left[\vec{V}\left(\hat{\beta}_2\right)\right]^{1/2}$ $\vec{E}\left(\hat{\beta}_2\right)$



0.3

true values: 0 0.3 0.6

Dependent Variable: CONS
Method: Least Squares
Sample: 2 80
Included observations: 79

Variable	Coefficient	Std. Error	t-Statistic	Prob.
C	-6.109104	3.652678	-1.672500	0.0985
INC	0.489518	0.041982	11.66010	0.0000
CONS(-1)	0.415420	0.053688	7.737671	0.0000

0.4

R-squared	0.901808	Mean dependent var	90.34697
Adjusted R-squared	0.899224	S.D. dependent var	1.166393
S.E. of regression	0.370274	Akaike info criterion	0.888086
Sum squared resid	10.41980	Schwarz criterion	0.978065
Log likelihood	-32.07941	F-statistic	348.9978
Durbin-Watson stat	1.162167	Prob(F-statistic)	0.000000

$\vec{E}\left\{\left[\hat{V}\left(\hat{\beta}_2\right)\right]^{1/2}\right\}$



Breusch-Godfrey Serial Correlation LM Test: F-statistic=17.47141 Prob. F(2,74)=0.000001
White Heteroskedasticity Test: F-statistic=1.670890 Prob. F(5,73)=0.152431

With 99.5% probability B2OLS has bias in CI: (0.527 - 0.3) ± 3 x 0.043/100 = [0.226 , 0.228]

$\left\{\vec{V}\left[\vec{E}\left(\hat{\beta}_2\right)\right]\right\}^{1/2}$

EQ2, TSLS structural form

$\left[\vec{V}\left(\hat{\beta}_2\right)\right]^{1/2}$ $\vec{E}\left(\hat{\beta}_2\right)$



0.3

true values: 0 0.3 0.6

Dependent Variable: CONS
Method: Two-Stage Least Squares
Sample: 2 80
Included observations: 79
Instrument list: C GOV INV CONS(-1)

Variable	Coefficient	Std. Error	t-Statistic	Prob.
C	2.990133	4.618518	0.647423	0.5193
INC	0.237702	0.062273	3.817100	0.0003
CONS(-1)	0.650092	0.073208	8.880054	0.0000

0.4

R-squared	0.855325	Mean dependent var	90.34697
Adjusted R-squared	0.851518	S.D. dependent var	1.166393
S.E. of regression	0.449451	Sum squared resid	15.35246
Durbin-Watson stat	1.801300	Second-stage SSR	26.11669

$\vec{E}\left\{\left[\hat{V}\left(\hat{\beta}_2\right)\right]^{1/2}\right\}$



With 99.5% probability B2TSLS has bias in CI: (0.321 - 0.3) ± 3 x 0.056/100 = [0.019 , 0.022]

$\left\{\vec{V}\left[\vec{E}\left(\hat{\beta}_2\right)\right]\right\}^{1/2}$

Actual type I error probability (α_T); nominal significance level $\alpha = 0.05$



	REJTRUEB2OLS	REJTRUEB2TSLS	REJTRUEB4ROLS
Mean	0.999700	0.115300	0.161700
Std. Dev.	0.017319	0.319400	0.368194
Sum	9997.000	1153.000	1617.000
Observ.	10000	10000	10000

With 99.5% probability α_T of the OLS test on γ_4 is in CI: 0.162 ± 3 x 0.368/100 = [0.151 , 0.173], which is much larger than the nominal level 0.05

Using OLS in the structural form: The probability to reject the true value is about 1!

With 99.5% probability α_T of the TSLS test on β_2 is in CI: 0.115 ± 3 x 0.319/100 = [0.106 , 0.124], which is substantially larger than the nominal level 0.05

Actual rejection probability of false null; $\alpha = 0.05$

	REJFALSEB2OLS	REJFALSEB2TSLS	REJFALSEB4ROLS
Mean	0.167100	0.957700	0.151400
Std. Dev.	0.373084	0.201283	0.358456
Sum	1671.000	9577.000	1514.000
Observ.	10000	10000	10000

H_0: $\beta_2 = 0.5$
H_1: $\beta_2 \neq 0.5$
true value 0.3

H_0: $\gamma_4 = 0.757$
H_1: $\gamma_4 \neq 0.757$
true value 0.857

In the reduced form:
With 99.5% probability the rejection probability of $\gamma_4 = 0.757$ is in CI: 0.151 ± 3 x 0.358/100 = [0.140 , 0.162], which does not differ much from the type I error probability.

Using OLS in the structural form:
With 99.5% probability the rejection probability of $\beta_2 = 0.5$ is in CI: 0.167 ± 3 x 0.373/100 = [0.156 , 0.178], so much lower than the probability to reject the true value $\beta_2 = 0.3$

Using TSLS in the structural form:
With 99.5% probability the rejection probability of $\beta_2 = 0.5$ is in CI: 0.958 ± 3 x 0.201/100 = [0.952 , 0.964], so very much higher than the probability to reject the true value $\beta_2 = 0.3$